Newsgroup: it.comp.aiuto
Titolo: PPT->TXT?? [urgente...]

E' possibile estrarre automaticamente da un documento PPT tutto il testo per
inserirlo in un file TXT o DOC o altro? Ho già provato con le funzioni di
Power Point per esportare in html ma mi fa un file per ogni finestra ed è
peggio che andar di notte.
Grazie a chi mi risponde
--Scully

 peggio che andar di notte.
ppt2txt, shareware, 
http://www.convertzone.com
Ma con una macro non dovrebbe essere complicato. Se vuoi ne discutiamo
su microsoft.public.it.office.powerpoint.
-- 
Tiziano "Lapalissiano" Marmiroli
Microsoft MVP - Office

 su microsoft.public.it.office.powerpoint.
Il problema è che il ppt che devo convertire è zeppo di caselle di testo.
Odio chi l'ha fatto.
Grazie cmq!
Ciao
--Scully

 Il problema è che il ppt che devo convertire è zeppo di caselle di testo.
E' un problema per te, non dovrebbe esserlo per la macro (sempre che
abbia capito bene, PowerPoint non lo conosco molto).
Menu Strumenti>Macro> Visual Basic Editor
Menu Inserisci>Modulo
Incollaci 
'---------------------
Sub ppt2txt()
'salva il testo della presentazione nel
'file c:\prova.txt
    Dim s As Shape
    Dim oSld As Slide
    Dim FileNum As Integer
    Dim sTesto As String
    FileNum = FreeFile
    Open "c:\prova.txt" For Append As #FileNum
    For Each oSld In ActivePresentation.Slides
        For Each s In oSld.Shapes
            If s.HasTextFrame Then
                With s.TextFrame
                    If .HasText Then sTesto = .TextRange.Text
                    Print #FileNum, sTesto
                End With
            End If
        Next s
    Next oSld
    Close #FileNum
End Sub
'---------------------
Torna a PowerPoint e dal menu Strumenti>Macro>Macro lancia la macro
ppt2txt.
Facci sapere se e come hai risolto, grazie.


 Facci sapere se e come hai risolto, grazie.
Tiziano, sei un genio del VBA. Se lo sapevo ti chiedevo di farmi la parte di
VBA nell'esame di CAD ;) Scherzo, ovviamente. Perň si poteva fare uno
scambio. VBA per Java? No, eh?
Grazie, č funzionata perfettamente. Immagino che fare una cosa del genere
buttando tutto in un documento Word conservando la formattazione del testo
sia invece molto piů complesso, vero? :)
Grazie
--Scully
http://fabuland.altervista.org/

 sia invece molto più complesso, vero? :)
Falso! Oggi non ne becchi una! :)
Prova così.
'---------------------------------
Sub ppt2doc()
'salva il testo della presentazione
'in un documento di Word
    Dim s As Shape
    Dim oSld As Slide
    Dim wordApp As Object
    Dim wordDoc As Object
    Dim wordRng As Object
    Set wordApp = CreateObject("Word.Application")
    wordApp.documents.Add
    Set wordDoc = wordApp.ActiveDocument
    Set wordRng = wordDoc.Range
    wordApp.Visible = True
    For Each oSld In ActivePresentation.Slides
        For Each s In oSld.Shapes
            If s.HasTextFrame Then
                With s.TextFrame
                    If .HasText Then
                        .TextRange.Copy
                        wordRng.Collapse Direction:=0
                        wordRng.InsertParagraphAfter
                        wordRng.Paste
                    End If
                End With
            End If
        Next s
    Next oSld
    Set wordApp = Nothing
End Sub
'---------------------------------


 '---------------------------------
grazie! Funge!
ciao!
--Scully